

Capítulo 5

Espacios vectoriales. Aplicaciones lineales. Diagonalización de matrices por semejanza.

5.1. Espacios vectoriales

5.1.1. Combinaciones lineales. Dependencia e independencia lineal. Cambio de base

En un espacio vectorial V , un conjunto de vectores $\{u_1, u_2, \dots, u_n\}$, es linealmente dependiente si existen escalares no todos nulos a_1, a_2, \dots, a_n tales que $a_1 u_1 + a_2 u_2 + \dots + a_n u_n = \vec{0}$. Por tanto, comprobar si un conjunto de vectores es linealmente dependiente o independiente se reduce a resolver un sistema de ecuaciones lineales homogéneas.

Por ejemplo, vamos a estudiar la dependencia ó independencia lineal del conjunto de vectores siguientes en \mathbb{R}^4 :

$$\left\{ u_1 := (3, -1, 4, 2), u_2 := (2, 4, 3, -3), u_3 := (1, 3, 2, 5), u_4 := (1, -5, 1, 5) \right\}.$$

Para ello introducimos los comandos siguientes:

```
(%ixx) u1:[3,-1,4,2]$ u2:[2,4,3,-3]$ u3:[1,3,2,5]$ u4:[1,-5,1,5]$
(%ixx) solve(a1*u1+a2*u2+a3*u3+a4*u4,[a1,a2,a3,a4]);
solve: dependent equations eliminated: (4)
(%oxx) [[a1=-%r1,a2=%r1,a3=0,a4=%r1]]
```

El resultado devuelto por Maxima nos dice que el sistema homogéneo propuesto es compatible indeterminado, habiendo eliminado la ecuación número 4 al ser combinación lineal de las restantes. Además las soluciones del sistema vienen escritas utilizando el parámetro **%r1**.

Si ejecutamos de nuevo tales comandos, da la misma respuesta, salvo que ahora el parámetro que aparece se llama **%r2**.

Al hacer el parámetro igual a 1 obtenemos la solución particular no trivial $a_1 = -1$, $a_2 = 1$, $a_3 = 0$, $a_4 = 1$. Por tanto los vectores dados son linealmente dependientes.

Observe que al resolver un sistema de ecuaciones cuyas partes derechas son cero, como en este caso, no tenemos que escribir los ceros que aparecen a la derecha del signo igual. Maxima presupone que tales valores son cero. Por ejemplo, para resolver la ecuación $(x - 1) \cdot (x - 2) = 0$, tenemos las siguientes alternativas:

```
(%ixx) solve((x-1)*(x-2),x);
(%ixx) solve((x-1)*(x-2)=0,x);
```

Como puede apreciar, en ambos casos se obtiene la misma respuesta.

Continuando con el problema propuesto al principio, el que los vectores dados sean linealmente dependientes, equivale a que alguno de tales vectores sea combinación lineal de los restantes. Eso podemos comprobarlo a partir de la solución particular no trivial que hemos obtenido. Por ejemplo, como a_1 no es cero, ello significa que u_1 se puede expresar como combinación lineal de u_2, u_3, u_4 . Concretamente,

$$u_1 = 1 \cdot u_2 + 0 \cdot u_3 + 1 \cdot u_4.$$

Por la misma razón u_2 se puede expresar como combinación lineal de u_1, u_3, u_4 , y u_4 se puede expresar como combinación lineal de u_1, u_2, u_3 . Sin embargo, u_3 no se puede expresar como combinación lineal de u_1, u_2, u_4 .

Otra forma de llegar a estas mismas conclusiones es mediante la siguiente secuencia de comandos:

```
(%ixx) solve(b2*u2+b3*u3+b4*u4-u1,[b2,b3,b4]);
      solve(c1*u1+c3*u3+c4*u4-u2,[c1,c3,c4]);
      solve(d1*u1+d2*u2+d4*u4-u3,[d1,d2,d4]);
      solve(e1*u1+e2*u2+e3*u3-u4,[e1,e2,e3]);
solve: dependent equations eliminated: (4)
(%oxx) [[b2=1,b3=0,b4=1]]
solve: dependent equations eliminated: (4)
(%oxx) [[c1=1,c3=0,c4=-1]]
(%oxx) []
solve: dependent equations eliminated: (4)
(%oxx) [[e1=1,e2=-1,e3=0]]
```

Vemos cómo el primero, el segundo y el cuarto son combinación lineal del resto, mientras que el tercero no lo es.

También, para saber si son linealmente dependientes o independientes, podemos calcular el rango de la matriz formada por los vectores.

```
(%ixx) A:matrix([3,-1,4,2],[2,4,3,-3],[1,3,2,5],[1,-5,1,5])$ rank(A);
(%oxx) 3
```

Lo que nos dice que hay tres vectores linealmente independientes, aunque no sabríamos cuáles. Pero teniendo en cuenta lo que hemos obtenido con los comandos iniciales, vemos que pueden ser primero, segundo y tercero, ó primero, tercero y cuarto, ó segundo, tercero y cuarto.

Tomamos ahora los vectores u_1, u_2 y u_3 , que sabemos que son linealmente independientes. Vamos a comprobar si el vector $v = (1, 2, 1, 3)$ es combinación lineal de estos tres vectores.

```
(%ixx) v:[1,2,1,3]$ solve(a1*u1+a2*u2+a3*u3-v,[a1,a2,a3]);
(%oxx) []
```

Ésto significa que v no es combinación lineal de los otros tres, y así tenemos que los vectores u_1, u_2, u_3, v son linealmente independientes. Al tratarse de cuatro vectores linealmente independientes en un espacio vectorial de dimensión cuatro, tenemos una base de \mathbb{R}^4 .

Vamos a trabajar ahora en el espacio vectorial $(\mathbb{Z}_5)^5$. Para eso, primero modificamos el valor de la variable `modulus`.

```
(%ixx) modulus:5$
```

Tomamos los vectores $u_1 = (1, 2, 0, 1, 1)$, $u_2 = (3, 1, 2, 1, 0)$, $u_3 = (1, 1, 0, 1, 2)$, $u_4 = (3, 2, 4, 4, 1)$.

```
(%ixx) u1:[1,2,0,1,1]$ u2:[3,1,2,1,0]$ u3:[1,1,0,1,2]$ u4:[3,2,4,4,1]$
```

Comprobamos si son linealmente dependientes o independientes.

```
(%ixx) solve(a1*u1+a2*u2+a3*u3+a4*u4,[a1,a2,a3,a4]);
```

```
solve: dependent equations eliminated: (4 5)
(%oxx) [[a1=2*r,a2=-2*r,a3=r,a4=r]]
```

Lo que nos dice que son linealmente dependientes, y que cualquiera es combinación lineal del resto. Nos quedamos entonces con u_1, u_2, u_3 . Vamos a ampliar el conjunto $\{u_1, u_2, u_3\}$ a una base de $(\mathbb{Z}_5)^5$. Elegimos otro vector u_4 .

```
(%ixx) u4:[1,1,2,0,1]$ solve(a1*u1+a2*u2+a3*u3-u4,[a1,a2,a3]);
(%oxx) []
```

Y como $u_4 = (1, 1, 2, 0, 1)$ no es combinación lineal de u_1, u_2, u_3 tenemos que $\{u_1, u_2, u_3, u_4\}$ son vectores linealmente independientes.

Probamos ahora con un quinto vector u_5 .

```
(%ixx) u5:[2,1,0,3,0]$ solve(a1*u1+a2*u2+a3*u3+a4*u4-u5,[a1,a2,a3,a4]);
(%oxx) []
```

Y, al igual que antes, $u_5 = (2, 1, 0, 3, 0)$ no es combinación lineal de u_1, u_2, u_3, u_4 . Por tanto, el conjunto de vectores $B = \{u_1, u_2, u_3, u_4, u_5\}$ es linealmente independiente.

Tenemos entonces que B es una base de $V = (\mathbb{Z}_5)^5$.

Los vectores pertenecientes a B están escritos respecto de la base canónica de V , que denotamos por B_c . Por tanto la matriz del cambio de base de B a B_c es la matriz $A(B, B_c)$ cuyas columnas son los vectores de B en la base canónica.

```
(%ixx) A21:transpose(matrix(u1,u2,u3,u4,u5))$
```

La matriz del cambio de base de B_c a B es la inversa de $A(B, B_c)$.

```
(%ixx) A12:rat(invert(A21));
```

```
(%oxx) /R/
```

$$\begin{bmatrix} 0 & 2 & 1 & 1 & 1 \\ -1 & 2 & 1 & 0 & 2 \\ 2 & 0 & 1 & 2 & 1 \\ 1 & -2 & 2 & 0 & -2 \\ -2 & 2 & -1 & 1 & 2 \end{bmatrix}$$

Hemos usado el comando `rat` para que reduzca el resultado módulo 5.

Supongamos que tenemos el vector v cuyas coordenadas en la base B son $(1, 2, 1, 1, 3)$. ¿De qué vector estamos hablando?

Para obtener v , podemos calcular $v = 1 \cdot u_1 + 2 \cdot u_2 + 1 \cdot u_3 + 1 \cdot u_4 + 3 \cdot u_5$ y usar las expresiones de u_1, \dots, u_5 . O bien, podemos multiplicar la matriz del cambio de base de B a B_c por el vector columna

$$\begin{pmatrix} 1 \\ 2 \\ 1 \\ 1 \\ 3 \end{pmatrix}.$$

```
(%ixx) c:transpose(matrix([1,2,1,1,3]))$
(%ixx) rat(u1+2*u2+u3+u4+3*u5); rat(A21.c);
(%oxx) /R/ [0,-1,1,-2,-1]
```

```
(%oxx) /R/
```

$$\begin{bmatrix} 0 \\ -1 \\ 1 \\ -2 \\ -1 \end{bmatrix}$$

El vector que resulta es $v = (0, -1, 1, -2, -1) = (0, 4, 1, 3, 4)$. Observe que aplicando el primer método, el vector resultante se obtiene en forma de fila, mientras que aplicando el segundo método se obtiene en forma de columna.

Si ahora quisiéramos calcular las coordenadas del vector $(1, 1, 1, 1, 1)$ en la base B tendríamos que multiplicar por la matriz del cambio de base de la base canónica B_c a B .

```
(%ixx) c:transpose(matrix([1,1,1,1,1]))$ rat(A12.c);
```

$$(\% \text{oxx}) / \mathbb{R} / \begin{bmatrix} 0 \\ -1 \\ 1 \\ -1 \\ 2 \end{bmatrix}$$

Lo que nos dice que $(1, 1, 1, 1, 1) = -u_2 + u_3 - u_4 + 2 \cdot u_5$.

También podríamos haberlo calculado

```
(%ixx) solve(a1*u1+a2*u2+a3*u3+a4*u4+a5*u5-[1,1,1,1,1],[a1,a2,a3,a4,a5]);
```

```
(%oxx) [[a1=0,a2=-1,a3=1,a4=-1,a5=2]]
```

Sean ahora los vectores $v_1 := (1, 2, 1, 3, 3)$, $v_2 := (2, 1, 4, 0, 1)$, $v_3 := (3, 3, 4, 0, 0)$, $v_4 := (0, 4, 2, 2, 1)$ y $v_5 := (1, 2, 1, 1, 4)$.

```
(%ixx) v1:[1,2,1,3,3]$ v2:[2,1,4,0,1]$ v3:[3,3,4,0,0]$
```

```
          v4:[0,4,2,2,1]$ v5:[1,2,1,1,4]$
```

El conjunto $B' = \{v_1, v_2, v_3, v_4, v_5\}$ es una base de $(\mathbb{Z}_5)^5$, pues

```
(%ixx) C:matrix(v1,v2,v3,v4,v5);
```

$$(\% \text{oxx}) / \mathbb{R} / \begin{bmatrix} 1 & 2 & 1 & 3 & 3 \\ 2 & 1 & 4 & 0 & 1 \\ 3 & 3 & 4 & 0 & 0 \\ 0 & 4 & 2 & 2 & 1 \\ 1 & 2 & 1 & 1 & 4 \end{bmatrix}$$

```
(%ixx) rank(C);
```

```
(%oxx) 5
```

Además C^t es la matriz del cambio de base de B' a la base canónica B_c .

El cálculo de la matriz del cambio de base de B' a B lo realizamos teniendo en cuenta que para cambiar coordenadas de B' a B , primero cambiamos coordenadas de B' a B_c , y a continuación cambiamos coordenadas de B_c a B . Escrito ésto en forma matricial resulta

$$A(B', B) = A(B_c, B) \cdot A(B', B_c).$$

Nótese que el miembro de la derecha de la igualdad anterior tenemos que leerlo de derecha a izquierda, es decir, de B' primero cambiamos a B_c y después cambiamos de B_c a B . Entonces:

```
(%ixx) ABprimaB:A12.transpose(C);
```

$$(\% \text{oxx}) / \mathbb{R} / \begin{bmatrix} 1 & 2 & 0 & -2 & 0 \\ 0 & 1 & 2 & 2 & 2 \\ 2 & -1 & 0 & 2 & -1 \\ -2 & 1 & 0 & -1 & 1 \\ 0 & 1 & 1 & 0 & 0 \end{bmatrix}$$

Claro, que también puede obtenerse teniendo en cuenta que la primera columna está formada por las coordenadas del vector v_1 en la base B . E igual con el resto de columnas.

```
(%ixx) solve(a1*u1+a2*u2+a3*u3+a4*u4+a5*u5-v1,[a1,a2,a3,a4,a5]);
```

```
(%oxx) [[a1=1,a2=0,a3=2,a4=-2,a5=0]]
```

Las coordenadas de v_2, v_3, v_4, v_5 se obtienen similarmente.

La matriz del cambio de base de B a B' es la inversa de la matriz del cambio de base de B' a B .

```
(%ixx) ABBprima:ABprimaB^(-1);
```

$$(\% \text{oxx}) /R/ \begin{bmatrix} -2 & 2 & -2 & -1 & 1 \\ -1 & -1 & 2 & -1 & 2 \\ 1 & 1 & -2 & 1 & -1 \\ 0 & 0 & 1 & 1 & 0 \\ 2 & 0 & 0 & 1 & 0 \end{bmatrix}$$

5.1.2. Subespacios vectoriales.

Si V es un espacio vectorial y U es un subconjunto suyo (no vacío), decimos que U es un subespacio vectorial si U es cerrado para la suma y para el producto por escalares.

Para dar un subespacio vectorial de V , podemos hacerlo básicamente de dos formas.

Una primera sería dando un sistema de generadores del subespacio. Para esto necesitamos dar un subconjunto S de V , y el subespacio sería el conjunto de todas las combinaciones lineales de los vectores de S .

Una segunda forma consistiría en establecer las condiciones que deben satisfacer los vectores de V para pertenecer al subespacio. (Realmente serían las coordenadas de estos vectores, con lo cual habría que referirlos a una base.) Estas ecuaciones tienen que ser lineales (de grado uno) y homogéneas (los términos independientes son nulos).

Vamos a tomar el subespacio de \mathbb{R}^4 generado por los vectores $(1, 2, 1, -1)$, $(2, 0, 0, -1)$, $(1, -1, -2, 1)$, $(0, 1, 3, -1)$. Llamemos a este subespacio U .

```
(%ixx) modulus:false$
```

```
(%ixx) u1:[1,2,1,-1]$ u2:[2,0,0,-1]$ u3:[1,-1,2,1]$ u4:[0,1,3,1]$
```

```
A:matrix(u1,u2,u3,u4);
```

$$(\% \text{oxx}) \begin{bmatrix} 1 & 2 & 1 & -1 \\ 2 & 0 & 0 & -1 \\ 1 & -1 & 2 & 1 \\ 0 & 1 & 3 & 1 \end{bmatrix}$$

Ahora, para obtener un vector del subespacio U nos basta con realizar una combinación lineal de los vectores u_1, u_2, u_3, u_4 . Por ejemplo: $v = 2 \cdot u_1 - 3 \cdot u_2 + u_3 - u_4 = (-3, 2, 1, 1)$.

```
(%ixx) v:2*u1-3*u2+u3-u4;
```

```
(%oxx) [-3,2,1,1]
```

O si queremos:

```
(%ixx) c:matrix([2,-3,1,-1])$ v:c.A;
```

```
(%oxx) [-3 2 1 1]
```

Para calcular la dimensión del subespacio que generan, no tenemos más que calcular el rango de la matriz cuyas filas son los vectores que generan el subespacio.

```
(%ixx) rank(A);
```

```
(%oxx) 3
```

Puesto que la dimensión es tres, y tenemos un sistema de generadores formado por cuatro vectores, éstos son linealmente dependientes. Vamos a comprobarlo.

```
(%ixx) solve(a1*u1+a2*u2+a3*u3+a4*u4,[a1,a2,a3,a4]);
```

```
solve: dependent equations eliminated: (4)
```

```
(%oxx) [[a1=-%r,a2=%r,a3=-%r,a4=%r]]
```

Lo que nos dice que $-u_1 + u_2 - u_3 + u_4 = (0, 0, 0, 0)$.

También podemos encontrar esta combinación lineal como sigue. A la matriz A le añadimos a la derecha la matriz identidad, y calculamos una forma escalonada de la matriz resultante. Entonces, las operaciones elementales que realizamos para llegar a la forma escalonada de A las tenemos "guardadas".

```
(%ixx) C:addcol(A,ident(4))$ D1:triangularize(C); D2:echelon(C);
```

```
(%oxx) [ 2  0  0 -1  0  1  0  0 ]
        [ 0  2  6  2  0  0  0  2 ]
        [ 0  0 10  5  0 -1  2  2 ]
        [ 0  0  0  0 10 -10 10 -10 ]
```

```
(%oxx) [ 1  0  0 -1/2  0  1/2  0  0 ]
        [ 0  1  3  1  0  0  0  1 ]
        [ 0  0  1  1/2  0 -1/10 1/5 1/5 ]
        [ 0  0  0  0  1 -1  1 -1 ]
```

Si nos fijamos, podemos ver que la última fila de la matriz A se ha transformado en la fila nula. Los otros cuatro elementos de esa fila nos dicen los coeficientes de la combinación lineal que nos da esa fila (nula). Según la primera matriz, $D1$, tenemos $10u_1 - 10u_2 + 10u_3 - 10u_4 = \vec{0}$, mientras que según la segunda matriz, $D2$, tenemos $u_1 - u_2 + u_3 - u_4 = \vec{0}$.

Para comprender mejor por qué ésto es así definimos las matrices siguientes:

```
(%ixx) P1:submatrix(D1,1,2,3,4)$ P2:submatrix(D2,1,2,3,4)$ P1.A; P2.A;
```

```
(%oxx) [ 2  0  0 -1 ]
        [ 0  2  6  2 ]
        [ 0  0 10  5 ]
        [ 0  0  0  0 ]
```

```
(%oxx) [ 1  0  0 -1/2 ]
        [ 0  1  3  1 ]
        [ 0  0  1  1/2 ]
        [ 0  0  0  0 ]
```

Vemos entonces que $P1$ es una matriz regular tal que $P1.A$ es el resultado del comando `triangularize(A)`, y $P2$ es una matriz regular tal que $P2.A$ es el resultado del comando `echelon(A)`. En el primer caso la cuarta fila de la matriz $P1.A$ es nula, lo cual significa que al formar una combinación lineal de las filas de A tomando como coeficientes los de la cuarta fila de $P1$ da la fila nula. Similarmente para $P2.A$.

Si ahora lo que queremos es comprobar si un vector pertenece o no al subespacio U , lo que tenemos que ver es si es combinación lineal de los vectores que generan a U . Por ejemplo, vamos a ver si los vectores $v1 = (7, 15, 10, -6)$ y $v2 = (5, -11, 10, 7)$ pertenecen ó no a U .

```
(%ixx) v1:[7,15,10,-6]$ solve(a1*u1+a2*u2+a3*u3+a4*u4-v1,[a1,a2,a3,a4]);
solve: dependent equations eliminated: (4)
(%oxx) [[a1=8-%r,a2=%r-1,a3=1-%r,a4=%r]]
(%ixx) v2:[5,-11,10,7]$ solve(a1*u1+a2*u2+a3*u3+a4*u4-v2,[a1,a2,a3,a4]);
(%oxx) []
```

Lo que nos dice que $v1 \in U$, pero $v2 \notin U$.

También podríamos haberlo comprobado estudiando los rangos de las matrices.

```
(%ixx) B1:addrow(A,v1)$ B2:addrow(A,v2)$ is(rank(A)=rank(B1)); is(rank(A)=rank(B2));
(%oxx) true
(%oxx) false
```

Es decir, $v1 \in U$, y $v2 \notin U$, como ya sabíamos.

Podemos utilizar esta idea para calcular unas ecuaciones implícitas de U .

En primer lugar, la dimensión del subespacio U es 3, y sabemos que el cuarto vector es combinación lineal del resto. Entonces, tomamos la matriz A que tiene como filas los tres primeros vectores.

```
(%ixx) A:matrix(u1,u2,u3)$
```

Si $v = (x, y, z, t)$ es un vector de \mathbb{R}^4 , para que pertenezca al subespacio U es necesario y suficiente que siga valiendo 3 el rango de la matriz que resulta de añadirle a la matriz A la fila (x, y, z, t) . Para que esto ocurra, el determinante de esa matriz tiene que valer cero.

```
(%ixx) v:[x,y,z,t]$ rat(determinant(addrow(A,v)));
```

```
(%oxx) /R/ 5z-5x-5y-10t
```

Luego los vectores de U cumplen todos la ecuación $5x + 5y - 5z + 10t = 0$, que dividiendo por 5 nos queda $x + y - z + 2t = 0$.

Supongamos ahora que U es el subespacio de \mathbb{R}^4 generado por los vectores $(2, 1, 3, -1)$, $(2, 4, 0, 2)$ y $(1, 1, 1, 0)$. Vamos a calcular unas ecuaciones cartesianas de U .

```
(%ixx) kill(all)$ u1:[2,1,3,-1]$ u2:[1,2,0,1]$ u3:[1,1,1,0]$
```

```
(%ixx) A:matrix(u1,u2,u3); rank(A);
```

```
(%oxx)  $\begin{bmatrix} 2 & 1 & 3 & -1 \\ 1 & 2 & 0 & 1 \\ 1 & 1 & 1 & 0 \end{bmatrix}$ 
```

```
(%oxx) 2
```

Luego la dimensión del subespacio vale 2. Al ser un subespacio de un espacio vectorial de dimensión 4, vendrá dado por $4 - 2 = 2$ ecuaciones cartesianas. Para encontrar esas dos ecuaciones, procedemos como sigue:

Calculamos una base de U :

```
(%ixx) B:echelon(A);
```

```
(%oxx)  $\begin{bmatrix} 1 & 2 & 0 & 1 \\ 0 & 1 & -1 & 1 \\ 0 & 0 & 0 & 0 \end{bmatrix}$ 
```

Luego una base del subespacio es $B = \{(1, 2, 0, 1); (0, 1, -1, 1)\}$. Formamos la matriz cuyas filas son los vectores de la base, y le añadimos el vector (x, y, z, t) .

```
(%ixx) v:[x,y,z,t]$ A:matrix(B[1],B[2],v);
```

```
(%oxx)  $\begin{bmatrix} 1 & 2 & 0 & 1 \\ 0 & 1 & -1 & 1 \\ x & y & z & t \end{bmatrix}$ 
```

Para que el vector (x, y, z, t) pertenezca al subespacio, es necesario que el rango de la matriz A valga 2, luego todos los determinantes 3×3 que podamos formar deben ser nulos. Para ello, partimos de un determinante no nulo 2×2 (el formado por las dos primeras filas y las dos primeras columnas), y le añadimos, las columnas tercera y cuarta.

```
(%ixx) eq1:determinant(submatrix(A,4)) = 0; eq2:determinant(submatrix(A,3))=0;
```

```
(%oxx) z+y-2x=0
```

```
(%oxx) -y+x+t=0
```

Que son las dos ecuaciones del subespacio U .

Si suponemos que no disponemos de la información anterior y ahora U viene definido por tales ecuaciones, resolviendo el sistema, podemos obtener unas ecuaciones paramétricas y una base de U .

```
(%ixx) linsolve([eq1,eq2],[x,y,z,t]);
```

```
(%oxx) [x=%r2,y=%r2+%r1,z=%r2-%r1,t=%r1]
```

Y una base la obtenemos dándole a r_1 y r_2 los valores 0, 1 y 1, 0 respectivamente.

5.2. Aplicaciones lineales

Recordemos que una aplicación lineal $f : V \rightarrow V'$ queda determinada al especificar las imágenes de los vectores de un sistema de generadores de V . En particular, f queda totalmente determinada al especificar las imágenes de los vectores de una base de V .

Sean los vectores de \mathbb{R}^4 siguientes:

```
(%ixx) kill(all)$ u1:[1,1,0,-1]$ u2:[0,2,-4,1]$ u3:[1,1,1,0]$ u4:[1,-1,1,3]$
```

Definimos

```
(%ixx) A:matrix(u1,u2,u3,u4)$
```

y calculamos

```
(%ixx) rank(A);
```

```
(%oxx) 4
```

Por tanto $\{u_1, u_2, u_3, u_4\}$ es una base de \mathbb{R}^4 .

Vamos a determinar la expresión de la única aplicación lineal $f : \mathbb{R}^4 \rightarrow \mathbb{R}^3$ que verifica

$$\begin{aligned}f(u_1) &= (0, 1, 2) \\f(u_2) &= (-4, 7, 1) \\f(u_3) &= (3, 0, 5) \\f(u_4) &= (0, 1, 2)\end{aligned}$$

Si escribimos

$$f(x, y, z, t) = (a_{11}x + a_{12}y + a_{13}z + a_{14}t, a_{21}x + a_{22}y + a_{23}z + a_{24}t, a_{31}x + a_{32}y + a_{33}z + a_{34}t),$$

las condiciones exigidas nos llevan a

$$\begin{pmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \end{pmatrix} \cdot \begin{pmatrix} 1 & 0 & 1 & 1 \\ 1 & 2 & 1 & -1 \\ 0 & -4 & 1 & 1 \\ -1 & 1 & 0 & 3 \end{pmatrix} = \begin{pmatrix} 0 & -4 & 3 & 0 \\ 1 & 7 & 0 & 1 \\ 2 & 1 & 5 & 2 \end{pmatrix}$$

de donde

$$\begin{pmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \end{pmatrix} = \begin{pmatrix} 0 & -4 & 3 & 0 \\ 1 & 7 & 0 & 1 \\ 2 & 1 & 5 & 2 \end{pmatrix} \cdot \begin{pmatrix} 1 & 0 & 1 & 1 \\ 1 & 2 & 1 & -1 \\ 0 & -4 & 1 & 1 \\ -1 & 1 & 0 & 3 \end{pmatrix}^{-1}.$$

La matriz de la derecha la calculamos fácilmente usando Maxima mediante los comandos siguientes:

```
(%ixx) Af:matrix([0,-4,3,0],[1,7,0,1],[2,1,5,2]).invert(transpose(A));
```

El resultado es

$$\begin{pmatrix} -\frac{29}{16} & \frac{39}{16} & \frac{19}{8} & \frac{5}{8} \\ \frac{5}{4} & \frac{1}{4} & -\frac{3}{2} & \frac{1}{2} \\ -\frac{1}{8} & \frac{27}{8} & \frac{7}{4} & \frac{5}{4} \end{pmatrix}.$$

Finalmente para obtener la expresión de f escribimos lo siguiente:

```
(%ixx) Af.matrix([x],[y],[z],[t]);
```

y obtenemos

$$\begin{pmatrix} \frac{19z}{8} + \frac{39y}{16} - \frac{29x}{16} + \frac{5t}{8} \\ -\frac{3z}{2} + \frac{y}{4} + \frac{5x}{4} + \frac{t}{2} \\ \frac{7z}{4} + \frac{27y}{8} - \frac{x}{8} + \frac{5t}{4} \end{pmatrix}.$$

Observamos que en su respuesta, Maxima ordena dentro de las expresiones aritméticas las variables x, y, z, t de otra forma. Para que las ordene de la forma usual añadimos:

```
(%ixx) ratvars(t,z,y,x)$
(%ixx) rat(Af.matrix([x],[y],[z],[t]));
```

Resulta

$$\begin{pmatrix} -\frac{29x-39y-38z-10t}{16} \\ \frac{5x+y-6z+2t}{4} \\ -\frac{x-27y-14z-10t}{8} \end{pmatrix},$$

con lo cual

$$f(x, y, z, t) = \left(-\frac{29x}{16} + \frac{39y}{16} + \frac{19z}{8} + \frac{5t}{8}, \frac{5x}{4} + \frac{y}{4} - \frac{3z}{2} + \frac{t}{2}, -\frac{x}{8} + \frac{27y}{8} + \frac{7z}{4} + \frac{5t}{4} \right).$$

Comprobamos ahora que la aplicación obtenida verifica las condiciones requeridas. Para ello comenzamos definiendo f .

```
(%ixx) f(v):=[-29*v[1]/16+39*v[2]/16+19*v[3]/8+5*v[4]/8,
5*v[1]/4+v[2]/4-3*v[3]/2+v[4]/2,
-v[1]/8+27*v[2]/8+7*v[3]/4+5*v[4]/4];
```

Finalmente calculamos

```
(%ixx) f(u1); f(u2); f(u3); f(u4);
(%oxx) [0,1,2]
(%oxx) [-4,7,1]
(%oxx) [3,0,5]
(%oxx) [0,1,2]
```

vectores que coinciden con los requeridos.

Ahora vamos a calcular el núcleo de la aplicación lineal f . Para ello introducimos:

```
(%ixx) solve([-29*x/16+39*y/16+19*z/8+5*t/8=0,
5*x/4+y/4-3*z/2+t/2=0,
-x/8+27*y/8+7*z/4+5*t/4=0], [x,y,z,t]);
```

Obtenemos

```
(%oxx) [[x=0,y=%r/2,z=%r/4,t=%r]]
```

con lo cual el núcleo de f , $Nuc(f)$, es el subespacio de \mathbb{R}^4 dado por las ecuaciones paramétricas

$$\begin{cases} x = 0 \\ y = -2\lambda \\ z = \lambda \\ t = 4\lambda \end{cases}$$

y $\{(0, -2, 1, 4)\}$ es una base del subespacio $Nuc(f)$. De hecho podemos hacer la siguiente comprobación:

```
(%ixx) f([0,-2*lamda,lamda,4*lamda]);
```

```
(%oxx) [0,0,0]
```

Como $Nuc(f)$ no es el subespacio trivial de \mathbb{R}^4 , deducimos que f no es inyectiva.

Con ésto, hemos calculado de forma artesanal el núcleo de una aplicación lineal. En la sección siguiente usaremos el comando `nullspace(A)` que calcula una base para el subespacio núcleo correspondiente a la aplicación lineal definida por la matriz A . Si escribes `nullspace(Af)`, puede observar que la respuesta no es muy agradable.

A continuación calculamos una base para el subespacio imagen de f , es decir, $Im(f)$. Por la teoría sabemos que

$$Im(f) = \langle f(u1), f(u2), f(u3), f(u4) \rangle.$$

Plasmamos ésto en Maxima:

```
(%ixx) AIm:matrix(f(u1),f(u2),f(u3),f(u4));
```

Obtenemos la matriz

$$\begin{pmatrix} 0 & 1 & 2 \\ -4 & 7 & 1 \\ 3 & 0 & 5 \\ 0 & 1 & 2 \end{pmatrix}.$$

cuyas filas representan un sistema de generadores de $Im(f)$. Para obtener una base para $Im(f)$, la triangularizamos:

```
(%ixx) BIm:triangularize(AIm);
```

Obtenemos la matriz

$$\begin{pmatrix} 3 & 0 & 5 \\ 0 & 21 & 23 \\ 0 & 0 & 19 \\ 0 & 0 & 0 \end{pmatrix}$$

Por tanto $\{(3, 0, 5), (0, 21, 23), (0, 0, 19)\}$ es una base de $Im(f)$. Como $Im(f)$ es subespacio de \mathbb{R}^3 , llegamos a que $Im(f) = \mathbb{R}^3$ y por tanto f es sobreyectiva.

Recordemos que si $f : V \rightarrow V'$ es lineal, con $\dim(V)$ finita, entonces $\dim(V) = \dim(Nuc(f)) + \dim(Im(f))$.

En el ejemplo anterior, una vez que sabemos que $\dim(Nuc(f)) = 1$, podríamos haber aplicado esta fórmula y haber obtenido que $\dim(Im(f)) = 4 - 1 = 3$, llegando a la misma conclusión, con menos esfuerzo.

La composición de aplicaciones lineales es de nuevo otra aplicación lineal. Concretamente, si $f : V \rightarrow V'$ y $g : V' \rightarrow V''$ son lineales, entonces la aplicación $g \circ f : V \rightarrow V''$ es lineal. Además, si B, B', B'' son bases de V, V', V'' , respectivamente, entonces

$$A(g \circ f, B, B'') = A(g, B', B'') \cdot A(f, B, B').$$

Ilustramos ésto con un ejemplo.

Sean $f : \mathbb{R}^4 \rightarrow \mathbb{R}^3$, $g : \mathbb{R}^3 \rightarrow \mathbb{R}^5$ dadas por $f(x, y, z, t) = (x + y - z, 4y - z + t, x + z + 5t)$ y $g(x, y, z) = (2x, 0, y + 6z, x - y + 8z, y - z)$.

Definimos estas aplicaciones en Maxima.

```
(%ixx) kill(all);
```

```
(%ixx) f(v):=[v[1]+v[2]-v[3], 4*v[2]-v[3]+v[4], v[1]+v[3]+5*v[4]];
```

```
(%ixx) g(v):=[2*v[1], 0, v[2]+6*v[3], v[1]-v[2]+8*v[3], v[2]-v[3]];
```

Hacemos la composición mediante

```
(%ixx) ratvars(t,z,y,x)$ rat(g(f([x,y,z,t])));
```

Nótese que el comando `rat` lo usamos para que simplifique la lista de expresiones aritméticas obtenidas.

Por tanto, si $h = g \circ f$, entonces $h : \mathbb{R}^4 \rightarrow \mathbb{R}^5$ y

$$h(x, y, z, t) = (2x + 2y - 2z, 0, 6x + 4y + 5z + 31t, 9x - 3y + 8z + 39t, -x + 4y - 2z - 4t).$$

Ahora obtendremos este mismo resultado mediante el producto de las matrices de f y g respecto de las bases canónicas.

```
(%ixx) Af:matrix([1,1,-1,0],[0,4,-1,1],[1,0,1,5]);
```

```
(%ixx) Ag:matrix([2,0,0],[0,0,0],[0,1,6],[1,-1,8],[0,1,-1]);
```

Entonces calculamos

```
(%ixx) Ah:Ag.Af;
```

siendo el resultado

$$\begin{pmatrix} 2 & 2 & -2 & 0 \\ 0 & 0 & 0 & 0 \\ 6 & 4 & 5 & 31 \\ 9 & -3 & 8 & 39 \\ -1 & 4 & -2 & -4 \end{pmatrix}$$

Ésta es la matriz de $h = g \circ f$ respecto de las bases canónicas.

La expresión de h respecto de las bases canónicas la obtenemos mediante

```
(%ixx) rat(Ah.matrix([x],[y],[z],[t]));
```

A partir de lo cual llegamos a la misma conclusión anterior.

5.3. Diagonalización

Dada una matriz cuadrada $A \in M_n(K)$, decimos que es diagonalizable (por semejanza) si existe una matriz regular P de forma que

$$P^{-1} \cdot A \cdot P$$

sea una matriz diagonal.

Por ejemplo, la matriz $A = \begin{pmatrix} 29 & -18 \\ 45 & -28 \end{pmatrix} \in M_2(\mathbb{R})$ es diagonalizable.

```
(%ixx) A:matrix([29,-18],[45,-28])$ P:matrix([2,3],[3,5])$ D:P^(-1).A.P;
```

```
(%oxx) [ 2  0 ]
       [ 0 -1 ]
```

Podemos constatar de modo “artesanal” que la matriz $N = \begin{pmatrix} 0 & 1 \\ 0 & 0 \end{pmatrix}$ no es diagonalizable. Para ello consideramos $Q = \begin{pmatrix} a & b \\ c & d \end{pmatrix}$ una matriz regular cualquiera.

```
(%ixx) N:matrix([0,1],[0,0])$ Q:matrix([a,b],[c,d])$ Q^(-1).N.Q;
```



```
(%oxx) [ cd/(ad-bc) d^2/(ad-bc) ]
       [ -c^2/(ad-bc) -cd/(ad-bc) ]
```

Para que la matriz así obtenida fuera diagonal tendría que ocurrir que tanto d^2 como c^2 fuesen cero, lo cual no es posible, pues en ese caso, la matriz Q no sería regular.

Caso de que una matriz cuadrada A sea diagonalizable, a la matriz P se le llama *matriz de paso*, y los elementos que aparecen en la diagonal principal de la matriz D son los *valores propios de la matriz A*. Estos valores propios se obtienen como las raíces del polinomio característico de A , es decir, el determinante de la matriz $A - x \cdot I$.

```
(%ixx) rat(determinant(A-x*ident(2)));
(%oxx) /R/ x^2-x-2
(%ixx) factor(%);
(%oxx) (x-2)(x+1)
```

Vemos entonces cómo las raíces del polinomio característico son $x = 2$ y $x = -1$, justamente los mismos elementos que nos aparecían en la matriz diagonal D .

Maxima dispone del comando `eigenvalues` para calcular los valores propios de una matriz.

```
(%ixx) eigenvalues(A);
(%oxx) [[2,-1],[1,1]]
```

Y el resultado es una lista constituida por dos sublistas. La primera está formada por los dos valores propios de la matriz A (es decir, 2 y -1). La segunda está formada por las multiplicidades algebraicas de estos valores propios, que en este caso son ambas 1. La multiplicidad algebraica de un valor propio a es el exponente al que está elevado el factor $(x - a)$ en la factorización del polinomio característico.

Por ejemplo, definimos $C = \begin{pmatrix} 2 & 0 & 0 \\ -1 & 2 & 1 \\ -1 & 0 & 3 \end{pmatrix}$.

```
(%ixx) C:matrix([2,0,0],[-1,2,1],[-1,0,3])$ pC:rat(determinant(C-x*ident(3)));
(%oxx) -x^3+7x^2-16x+12
(%ixx) factor(pC);
(%oxx) -(x-3)(x-2)^2
```

Lo que nos dice que tenemos dos valores propios $x = 3$ y $x = 2$ con multiplicidades algebraicas iguales a 1 y a 2, respectivamente. Ésto mismo lo comprobamos con el comando `eigenvalues`.

```
(%ixx) eigenvalues(C);
(%oxx) [[2,3],[2,1]]
```

Si x es un valor propio de una matriz A , y v es un vector columna con al menos una componente no nula, se dice que v es un vector propio de valor propio x si $A \cdot v = x \cdot v$. Por ejemplo, $v_1 = \begin{pmatrix} 2 \\ 3 \end{pmatrix}$ es un vector propio de A de valor propio $x = 2$.

```
(%ixx) v1:transpose(matrix([2,3]))$ A.v1;
(%oxx)  $\begin{bmatrix} 4 \\ 6 \end{bmatrix}$ 
```

Que como vemos es $2 \cdot v_1$.

Y $v_2 = \begin{pmatrix} 3 \\ 5 \end{pmatrix}$ es un vector propio de valor propio -1 .

```
(%ixx) v2:transpose(matrix([3,5]))$ A.v2;
(%oxx)  $\begin{bmatrix} -3 \\ -5 \end{bmatrix}$ 
```

Que es igual a $-v_2$.

Los vectores propios de valor propio x pueden calcularse como los vectores v tales que $(A - x \cdot I) \cdot v = 0$, es decir, los vectores del núcleo de la aplicación lineal dada por la matriz $A - x \cdot I$. Maxima dispone para esto del comando `nullspace`. Al conjunto de todos los vectores propios de valor propio x se le conoce como *subespacio propio* de valor propio x .

```
(%oxx) nullspace(A-2*ident(2)); nullspace(A+ident(2));  
(%oxx) span ([[ 18 ]]  
              [ 27 ]]  
(%oxx) span ([[ 18 ]]  
              [ 30 ]]
```

Lo que nos dice que ambos subespacios son los subespacios generados por los vectores $\begin{pmatrix} 18 \\ 27 \end{pmatrix}$ (eso es lo que significa *span*) y $\begin{pmatrix} 18 \\ 30 \end{pmatrix}$, respectivamente.

Pero también podemos hallar estos vectores con el comando `eigenvectors`.

```
(%ixx) eigenvectors(A);  
(%oxx) [[ [2,-1], [1,1] ], [[ [1, 3/2], [1, 5/3] ] ]]
```

Que como vemos nos devuelve una lista formada por dos sublistas. La primera sublista nos dice cuales son los valores propios y su multiplicidad algebraica (tal y como lo obteníamos con el comando `eigenvalues`), y la segunda sublista está formada por tantas listas como valores propios. Y cada una de ellas nos da una base del correspondiente subespacio propio.

Entonces, la información que nos proporciona `eigenvectors(A)` es:

La matriz A tiene dos valores propios, que son $x = 2$ y $x = -1$.

La multiplicidad algebraica del valor propio $x = 2$ vale 1, y la multiplicidad algebraica del valor propio $x = -1$ vale también 1.

Una base del subespacio propio de valor propio 2 es $\left\{ \begin{pmatrix} 1 \\ 3/2 \end{pmatrix} \right\}$ (podríamos multiplicar ese vector por 2, para tener sus coordenadas enteras).

Una base del subespacio propio de valor propio -1 es $\left\{ \begin{pmatrix} 1 \\ 5/3 \end{pmatrix} \right\}$.

Con los vectores que nos han salido, podemos formar una matriz regular P , poniendo esos vectores como columnas. Entonces, si $P = \begin{pmatrix} 1 & 1 \\ 3/2 & 5/3 \end{pmatrix}$, o si preferimos, la matriz $\begin{pmatrix} 2 & 3 \\ 3 & 5 \end{pmatrix}$ que resulta de multiplicar los dos vectores propios por 2 y por 3 respectivamente, se tiene que $P^{-1} \cdot A \cdot P$ es una matriz diagonal, como ya vimos al comienzo de la sesión.

Vamos a hacer lo mismo ahora con la matriz C .

```
(%ixx) vp:eigenvectors(C);  
(%oxx) [[ [2,3], [2,1] ], [[ [1,0,1], [0,1,0] ], [ [0,1,1] ] ]]
```

Y ahora tenemos:

La matriz C tiene dos valores propios, que son $x = 2$ y $x = 3$.

El valor propio 2 tiene multiplicidad algebraica 2, mientras que el valor propio 3 tiene multiplicidad algebraica 1.

El subespacio propio de valor propio 2 tiene dimensión 2, y una base es $\left\{ \begin{pmatrix} 1 \\ 0 \\ 1 \end{pmatrix}, \begin{pmatrix} 0 \\ 1 \\ 0 \end{pmatrix} \right\}$.

El subespacio propio de valor propio 3 tiene dimensión 1, y una base es $\left\{ \begin{pmatrix} 0 \\ 1 \\ 1 \end{pmatrix} \right\}$.

La matriz $Q = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 1 \\ 1 & 0 & 1 \end{pmatrix}$ es regular, y $Q^{-1} \cdot C \cdot Q$ es una matriz diagonal.

```
(%ixx) Q:matrix([1,0,0],[0,1,1],[1,0,1])$ Q^(-1).C.Q;
```

```
(%oxx)  $\begin{bmatrix} 2 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}$ 
```

Y vemos que la diagonal está formada justamente por los valores propios.

También podríamos haber obtenido la matriz Q como sigue:

```
(%ixx) Q:transpose(matrix(vp[2][1][1],vp[2][1][2],vp[2][2][1]));
```

```
(%oxx)  $\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 1 \\ 1 & 0 & 1 \end{bmatrix}$ 
```

Si cambiáramos el orden de las columnas, nos cambia el orden de los elementos de la diagonal.

```
(%ixx) Q:matrix([1,0,0],[0,1,1],[1,1,0])$ Q^(-1).C.Q;
```

```
(%oxx)  $\begin{bmatrix} 2 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 2 \end{bmatrix}$ 
```

Para que una matriz sea diagonalizable hace falta que la suma de las dimensiones de los subespacios propios sea igual al tamaño de la matriz.

Por ejemplo, sea $E = \begin{pmatrix} 0 & 0 & 1 \\ -2 & -1 & 2 \\ -1 & 0 & 2 \end{pmatrix}$. Entonces:

```
(%ixx) E:matrix([0,0,1],[-2,-1,2],[-1,0,2])$ eigenvectors(E);
```

```
(%oxx) [[[-1,1],[1,2]],[[[0,1,0]],[[1,0,1]]]]
```

Y vemos que:

La matriz E tiene dos valores propios, $x = -1$ y $x = 1$.

La multiplicidad algebraica de $x = -1$ vale 1, mientras que la multiplicidad algebraica de $x = 1$ vale 2.

El subespacio propio de valor propio $x = -1$ tiene dimensión 1, y una base es $\left\{ \begin{pmatrix} 0 \\ 1 \\ 0 \end{pmatrix} \right\}$.

El subespacio propio de valor propio $x = 1$ tiene dimensión 1, y una base es $\left\{ \begin{pmatrix} 1 \\ 0 \\ 1 \end{pmatrix} \right\}$.

La suma de las dimensiones de los subespacios propios vale 2. La matriz por tanto no es diagonalizable.

El hecho de que una matriz sea diagonalizable permite calcular de forma rápida potencias suyas, pues si

$$P^{-1} \cdot A \cdot P = D,$$

entonces

$$A = P \cdot D \cdot P^{-1},$$

luego

$$A^n = (P \cdot D \cdot P^{-1})^n = P \cdot D^n \cdot P^{-1},$$

y las potencias de una matriz diagonal se pueden calcular sin más que calcular las potencias de los elementos de la diagonal.

```
(%ixx) A^(15); D1:D^(15); P.D1.P^(-1);
```

```
(%oxx) [ 327689 -196614 ]
        [ 491535 -294922 ]
```

```
(%oxx) [ 32768  0 ]
        [  0   -1 ]
```

```
(%oxx) [ 327689 -196614 ]
        [ 491535 -294922 ]
```

Por supuesto, que todo esto podemos hacerlo módulo un número primo. Por ejemplo, vamos a trabajar módulo 7 con la matriz

$$A = \begin{pmatrix} 4 & 0 & 0 & 2 & 6 & 1 & 2 \\ 2 & 5 & 4 & 6 & 5 & 6 & 5 \\ 4 & 6 & 2 & 0 & 0 & 4 & 0 \\ 3 & 2 & 5 & 1 & 3 & 3 & 3 \\ 6 & 2 & 3 & 4 & 5 & 1 & 0 \\ 6 & 1 & 1 & 0 & 0 & 5 & 6 \\ 2 & 1 & 5 & 0 & 4 & 1 & 1 \end{pmatrix}$$

```
(%ixx) modulus:7$
```

```
(%ixx) A:matrix([4,0,0,2,6,1,2],[2,5,4,6,5,6,5],[4,6,2,0,0,4,0],[3,2,5,1,3,3,3],
                [6,2,3,4,5,1,0],[6,1,1,0,0,5,6],[2,1,5,0,4,1,1])$
```

```
(%ixx) vp:eigenvectors(A);
```

```
(%oxx) [[ [1,3,2,-2],[1,1,2,3]], [[ [1,2,-3,-3,-3,2,-1]],
    [[1,-1,1,-1,-1,-1,-3]], [[1,0,3,1,1,-1,-1],
```

```
(%oxx) [0,1,2,0,-1,2,2]], [[1,0,0,-3,-3,-1,-1],[0,1,0,-1,2,2,1],[0,0,1,3,0,-1,1]]]
```

Luego A tiene cuatro valores propios, que son $x = 1$, $x = 3$, $x = 2$ y $x = -2$, cuyas multiplicidades algebraicas son respectivamente 1, 1, 2 y 3.

Tenemos las siguientes bases para los subespacios propios:

$$B_{V_1} = \left\{ \begin{pmatrix} 1 \\ 2 \\ -3 \\ -3 \\ -3 \\ 2 \\ -1 \end{pmatrix} \right\}; \quad B_{V_3} = \left\{ \begin{pmatrix} 1 \\ -1 \\ 1 \\ -1 \\ -1 \\ -1 \\ -3 \end{pmatrix} \right\}; \quad B_{V_2} = \left\{ \begin{pmatrix} 1 \\ 0 \\ 3 \\ 1 \\ 1 \\ -1 \\ -1 \end{pmatrix}, \begin{pmatrix} 0 \\ 1 \\ 2 \\ 0 \\ -1 \\ 2 \\ 2 \end{pmatrix} \right\};$$

$$B_{V_{-2}} = \left\{ \begin{pmatrix} 1 \\ 0 \\ 0 \\ -3 \\ -3 \\ -1 \\ -1 \end{pmatrix}, \begin{pmatrix} 0 \\ 1 \\ 0 \\ -1 \\ 2 \\ 2 \\ 1 \end{pmatrix}, \begin{pmatrix} 0 \\ 0 \\ 1 \\ 3 \\ 0 \\ -1 \\ 1 \end{pmatrix} \right\}$$

Como la suma de las dimensiones de los subespacios propios vale 7, la matriz es diagonalizable. Y se tiene:


```
(%ixx) P:transpose(matrix(vp[2][1][1],vp[2][2][1],vp[2][3][1],vp[2][3][2],vp[2][4][1],
vp[2][4][2],vp[2][4][3]))$
(%ixx) rat(P^(-1).A.P);
(%oxx) 
$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 3 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 2 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -2 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -2 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -2 \end{bmatrix}$$

```

Vamos a tomar ahora la matriz

$$C = \begin{pmatrix} 4 & 4 & 6 & 1 & 5 & 2 & 6 \\ 6 & 3 & 2 & 5 & 5 & 4 & 2 \\ 2 & 2 & 3 & 6 & 0 & 2 & 0 \\ 4 & 2 & 2 & 0 & 3 & 1 & 4 \\ 4 & 0 & 5 & 2 & 6 & 6 & 6 \\ 5 & 5 & 2 & 5 & 6 & 1 & 0 \\ 5 & 4 & 4 & 4 & 4 & 1 & 6 \end{pmatrix}$$

Y vamos a ver si es o no diagonalizable.

```
(%ixx) C:matrix([-3,-3,-1,1,-2,2,-1],[-1,3,2,-2,-2,-3,2],[2,2,3,-1,0,2,0],[-3,2,2,0,3,1,-3],
[-3,0,-2,2,-1,-1,-1],[-2,-2,2,-2,-1,1,0],[-2,-3,-3,-3,-3,1,-1])$
(%ixx) eigenvectors(C); 
(%oxx) [[1,3,2,-2],[1,1,2,3]],[[1,-2,1,2,0,1,-2]],[[1,-2,-3,-1,-3,-3,2]],[[1,0,0,-3,3,1,2],
[0,1,-1,-1,-1,-1,-3]],[[1,0,-2,3,1,2,-1],[0,1,0,-1,-1,2,2]]]
```

Observamos que tiene los mismos valores propios que A , y con las mismas multiplicidades algebraicas. Pero la dimensión del subespacio propio de valor propio -2 tiene dimensión 2. La suma de las dimensiones de los subespacios propios es ahora igual a 6, luego la matriz C no es diagonalizable.

Ejercicio. Resuelva ejercicios de la Relación del Tema 5 con la ayuda de Maxima.